
Quickstart
    

Mix.install(
  [
    {:instructor, path: Path.expand("../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Introduction


Instructor is a library to do structured prompting with OpenAI and open source LLMs. While the idea is pretty simple, through this and the other examples you'll realize how powerful a concept this is.
So first off, what is structure prompting?
What if the LLM returned data conforming to a complicated nested schema that your code knows how to work with? Well, that's structure prompting. It's a way of cohercing the LLM to producing it's response in a known format that your downstream code can handle. In the case of Instructor, we use Ecto to provide those schemas. Good old Ecto, something you're already familiar with.
So, without further ado, let's take define a schema and take it for a spin!
defmodule Politician do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  A description of United States Politicians and the offices that they held,

  ## Fields:
  - first_name: Their first name
  - last_name: Their last name
  - offices_held:
    - office: The name of the political office held by the politician (in lowercase)
    - from_date: When they entered office (YYYY-MM-DD)
    - to_date: The date they left office, if relevant (YYYY-MM-DD or null).
  """
  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)

    embeds_many :offices_held, Office, primary_key: false do
      field(:office, Ecto.Enum,
        values: [:president, :vice_president, :governor, :congress, :senate]
      )

      field(:from_date, :date)
      field(:to_date, :date)
    end
  end
end
{:module, Politician, <<70, 79, 82, 49, 0, 0, 17, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
Great, we have our schema describing politicans and the offices they held. Let's notice a few things that may stand out from regular Ecto usage. First, since there is no database backing the schema, it doesn't make sense to give it a primary_key. This also makes sense because there is no sensible value for the LLM to respond with.
Also we use a @doc on the schema. This isn't just for documentation purposes of the tutorial. Instructor will take any @doc tag and provide it to the LLM. Generally you'll want to use this to provide semantic descriptions of the fields and general context to the LLM to ensure you get the outputs you want. In our case we want to push the LLM to understand that we are only considering American politicians.
So, let's try asking the LLM to give us some politicians.
Instructor.chat_completion(
  model: "gpt-4o-mini",
  response_model: Politician,
  messages: [
    %{
      role: "user",
      content:
        "Who won the American 2020 election and what offices have they held over their career?"
    }
  ]
)
{:ok,
 %Politician{
   first_name: "Joe",
   last_name: "Biden",
   offices_held: [
     %Politician.Office{office: :president, from_date: ~D[2021-01-20], to_date: nil},
     %Politician.Office{office: :vice_president, from_date: ~D[2009-01-20], to_date: ~D[2017-01-20]},
     %Politician.Office{office: :senate, from_date: ~D[1973-01-03], to_date: ~D[2009-01-15]}
   ]
 }}
Amazing, right? Using nothing more than one of the top libraries in Elixir, Ecto, we were able to get structured output from our LLM. The data returned is ready to be processed by our regular Elixir code. Instructor supports all field types that you can express in Ecto, including embedded and associated schemas.
It's almost as if the LLM inputted the data into a Phoenix Form. All the utilities that you use to process that kind of data, you can use to process the outputs of Instructor.
One of the superpowers of this is that since we're just using changesets under the hood, you can use the same validations that you would use elsewhere in your app. Let's look at that in the next section.

  
    
  
  Validations


Instructor provides a lightweight behavior where you can define a callback function that we will call to validate the data returned by the LLM using Ecto changesets. There is nothing fancy to this API. It's just a changeset in and a changeset out.
defmodule NumberSeries do
  use Ecto.Schema
  use Instructor.Validator

  @primary_key false
  embedded_schema do
    field(:series, {:array, :integer})
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_length(:series, min: 10)
    |> Ecto.Changeset.validate_change(:series, fn
      field, values ->
        if Enum.sum(values) |> rem(2) == 0 do
          []
        else
          [{field, "The sum of the series must be even"}]
        end
    end)
  end
end
{:module, NumberSeries, <<70, 79, 82, 49, 0, 0, 18, ...>>, {:validate_changeset, 1}}
In this albeit contrived example, we're going to get the LLM to return a series of numbers and validate whether it has at least 10 numbers and that the sum of the series is even.
When we ask for fewer than ten numbers, Instructor will return an error tuple with a change set that is invalid.
{:error, changeset} =
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    response_model: NumberSeries,
    messages: [
      %{role: "user", content: "Give me the first 5 integers"}
    ]
  )

# Render our the errors down to strings.
errors =
  Ecto.Changeset.traverse_errors(changeset, fn {msg, opts} ->
    Regex.replace(~r"%{(\w+)}", msg, fn _, key ->
      opts |> Keyword.get(String.to_existing_atom(key), key) |> to_string()
    end)
  end)

{changeset.changes, errors}
{%{series: [1, 2, 3, 4, 5]},
 %{series: ["The sum of the series must be even", "should have at least 10 item(s)"]}}
Now the beauty of this is that since we have human readable errors from our validations, we can just turn around and pass those back into the LLM to get it to fix its own errors.
Instructor provides a convenience parameter, max_retries for you in the initial call which will retry against the validations up to n times.
Instructor.chat_completion(
  model: "gpt-4o-mini",
  response_model: NumberSeries,
  max_retries: 10,
  messages: [
    %{role: "user", content: "Give some random integers"}
  ]
)

10:30:03.764 [debug] Retrying LLM call for NumberSeries:

 "series - The sum of the series must be even\nseries - should have at least 10 item(s)"

10:30:04.794 [debug] Retrying LLM call for NumberSeries:

 "series - The sum of the series must be even"

{:ok,
 %NumberSeries{series: [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20]}}
Here we demonstrated using regular Lixar code to validate the outputs of an LLM, but we don't have to stop there. We can actually use the LLM to validate the outputs of the LLM.
In Instructor, we have provided a custom Ecto Changset validator called validate_with_llm. 
Under the hood it just uses an instructor itself to check whether the field matches some condition that you have defined in plain text.
defmodule QuestionAnswer do
  use Ecto.Schema
  use Instructor.Validator

  @primary_key false
  embedded_schema do
    field(:question, :string)
    field(:answer, :string)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> validate_with_llm(:answer, "Do not say anything objectionable")
  end
end
{:module, QuestionAnswer, <<70, 79, 82, 49, 0, 0, 16, ...>>, {:validate_changeset, 1}}
%QuestionAnswer{}
|> Instructor.cast_all(%{
  question: "What is the meaning of life?",
  answer: "Sex, drugs, and rock'n roll"
})
|> QuestionAnswer.validate_changeset()
#Ecto.Changeset<
  action: nil,
  changes: %{question: "What is the meaning of life?", answer: "Sex, drugs, and rock'n roll"},
  errors: [answer: {"is invalid, Do not say anything objectionable", []}],
  data: #QuestionAnswer<>,
  valid?: false
>

  
    
  
  Record Streaming


Now if you've used chatGPT's web interface, you know that these LLMs can stream responses one token at a time. You can imagine that this is pretty easy to implement in code as you just reduce across the stream accumulating the value and appending to the log. But doing this when you're using OpenAI's function calls, and where the data is structured in JSON, it's not trivial to implement streaming.
Luckily we've done that work for you and we support two types of streaming in Instructor. The first is record streaming and the second is partial streaming.
First let's take a look at record streaming.
Record streaming is useful when you're asking the LLM for something that is naturally represented as an array of records. In this mode, we will instead of returning the full array, we'll return a stream that will emit each record once it's been completely streamed to the client, but before the next records tokens have arrived.
For example, let's take our presidents example from earlier, and we can instead ask for the first 5 presidents of the United States streaming each result as they come in.
presidents_stream =
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    stream: true,
    response_model: {:array, Politician},
    messages: [
      %{role: "user", content: "Who were the first 5 presidents of the United States?"}
    ]
  )
#Stream<[
  enum: #Function<60.53678557/2 in Stream.transform/3>,
  funs: [#Function<48.53678557/1 in Stream.map/2>]
]>
As you can see, instead of returning the result, we return a stream which can be run to emit each of the presidents.
presidents_stream
|> Stream.each(fn
  {:ok, politician} -> IO.inspect(politician)
  {:error, changeset} -> IO.inspect(changeset)
end)
|> Stream.run()
%Politician{
  first_name: "George",
  last_name: "Washington",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1789-04-30],
      to_date: ~D[1797-03-04]
    }
  ]
}
%Politician{
  first_name: "John",
  last_name: "Adams",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1797-03-04],
      to_date: ~D[1801-03-04]
    }
  ]
}
%Politician{
  first_name: "Thomas",
  last_name: "Jefferson",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1801-03-04],
      to_date: ~D[1809-03-04]
    }
  ]
}
%Politician{
  first_name: "James",
  last_name: "Madison",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1809-03-04],
      to_date: ~D[1817-03-04]
    }
  ]
}
%Politician{
  first_name: "James",
  last_name: "Monroe",
  offices_held: [
    %Politician.Office{
      office: :president,
      from_date: ~D[1817-03-04],
      to_date: ~D[1825-03-04]
    }
  ]
}
:ok
An important thing to note here is that we're running the validations independently for each value in the array.  That's why the values in the stream are either {:ok, Ecto.Schema.t()} or {:error, Ecto.Changeset.t()}.
As a result, it's unclear how we can automatically do retries to fix validation errors. And therefore, when in streaming mode, it is the responsibility of the user to retry when validation errors occur. (We may revisit this decision in the future)

  
    
  
  Partial Streaming


The other streaming mode that we have an instructor is called partial streaming. In this mode, you can get back a stream that will emit the record multiple times with the fields updating as they arrive. This can be used with a schema or an array of schemas. Both are demonstrated below.
This is useful in UI applications where you want to show instant feedback to the user about what data is showing up when without giving just some indeterminant loading spinner.
Instructor.chat_completion(
  model: "gpt-4o-mini",
  stream: true,
  response_model: {:partial, Politician},
  messages: [
    %{role: "user", content: "Who is the first president of the United States?"}
  ]
)
|> Stream.each(fn
  {:partial, politician} -> IO.puts("[Partial]: #{inspect(politician)}")
  {:ok, politician} -> IO.puts("[Final]: #{inspect(politician)}")
end)
|> Stream.run()
[Partial]: %Politician{first_name: nil, last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: nil, last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: nil, offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: []}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: nil, from_date: nil, to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: nil, from_date: ~D[1789-04-30], to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: nil}]}
[Partial]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: ~D[1797-03-04]}]}
[Final]: %Politician{first_name: "George", last_name: "Washington", offices_held: [%Politician.Office{office: :president, from_date: ~D[1789-04-30], to_date: ~D[1797-03-04]}]}
:ok
There is an important difference in this mode. Since your validations will be defined on the entirety of the object, it doesn't make sense to call the validate function until the entire record has been streamed in.
Therefore, we introduce a new output tuple in the stream compared to regular record streaming. The value can be {:partial, Ecto.Schema.t()}, and then on the last emit of the stream it can be {:error, Ecto.Changeset.t()}, or {:ok, Ecto.Schema.t()}
Like record streaming, however, using max_retries with this streaming mode does nothing and throws an error. (We may revisit this in the future when it's clear what such a behavior should do)

  
    
  
  Custom Ecto Types


Instructor supports all the Ecto types out of the box, but sometimes you need more. And that's why Instructor provides a behavior that you can implement on your own custom Ecto types. All you have to do is implement to_json_schema/0.
Whatever you return from this function will be put as the field type. See the JSONSchema Specification for more information on what you can put here. Typically you'll see people put description, type, and maybe format.
defmodule EctoURI do
  use Ecto.Type
  use Instructor.EctoType

  def type, do: :map

  # This is it, the rest is for implementing a regular old ecto type.
  def to_json_schema() do
    %{
      type: "string",
      description: "A valid URL"
    }
  end

  def cast(uri) when is_binary(uri) do
    {:ok, URI.parse(uri)}
  end

  def cast(%URI{} = uri), do: {:ok, uri}
  def cast(_), do: :error

  def load(data) when is_map(data) do
    data =
      for {key, val} <- data do
        {String.to_existing_atom(key), val}
      end

    {:ok, struct!(URI, data)}
  end

  def dump(%URI{} = uri), do: {:ok, Map.from_struct(uri)}
  def dump(_), do: :error
end
{:module, EctoURI, <<70, 79, 82, 49, 0, 0, 14, ...>>, {:dump, 1}}
Instructor.chat_completion(
  model: "gpt-4o-mini",
  response_model: %{url: EctoURI},
  messages: [
    %{role: "user", content: "Give me the URL for Google"}
  ]
)
{:ok,
 %{
   url: %URI{
     scheme: "https",
     authority: "www.google.com",
     userinfo: nil,
     host: "www.google.com",
     port: 443,
     path: nil,
     query: nil,
     fragment: nil
   }
 }}
And just like that, you can extend Instructor to get the LLM to return whatever you want.

Local Instructor w/ llama.cpp
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino_shell, "~> 0.1.2"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.Llamacpp,
      llamacpp: [
        chat_template: :mistral_instruct
      ]
    ]
  ]
)

  
    
  
  Setting up llama.cpp


llama.cpp is a great way to run models locally. Head on over to the repo and install it on your system.
Next, we'll need to download a GGUF compatible model to run with llama.cpp. As of today, I recommend using qwen-2.5-7b. It's a great model that is small enough to run locally.
A note on quantization: When you go search for GGUF models you'll see a lot of suffixes like Q4_K_M and F8. These are just different compression techniques, called quantization, that allow the model to take up dramatically less memory at the cost of some accuracy. There are many different methods of quantization that have different performance tradeoffs. However, it's generally recommended to run the largest model you can fit into your GPUs VRAM. Going over FP8 is generally unnecessary and at that point, you should be considering models with a larger number of parameters.

To start the llama server, run llama-server --port 8080 -ngl 999 -hf Qwen/Qwen2.5-7B-Instruct-GGUF. This will automatically download the model weights and start a server to run them. -ngl 999 is just a flag to tell llama how many layers of the neural network to offload to the GPU. 999, is effectively saying, run the entire model on the GPU.
Then with that running in the background. You can use Instructor as you normally would!
There are three ways to configure Instructor to use llama.cpp,
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.Llamacpp, llamacpp: [...]]])
	via config :instructor, adapter: Instructor.Adapters.Ollama, llamacpp: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime
config = [
  adapter: Instructor.Adapters.Llamacpp
]

defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    response_model: President,
    mode: :json_schema,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)
{:ok,
 %President{first_name: "George", last_name: "Washington", entered_office_date: ~D[1789-04-01]}}
And there you have it. You're running Instructor against a locally running large language model.  At zero incremental cost to you.

Local Instructor w/ Ollama
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ]
)

  
    
  
  Introduction


Before running the code below, please ensure the following:
	Pull the Required Model:
Run the following command to pull the necessary model image:
ollama pull qwen2.5:7b

This command downloads the model qwen2.5:7b so that it can be used by the Instructor adapter.

	Start the server:
Run the following command to start the local ollama server:
ollama serve



Once you have the server running, you can call it with Instructor using the Ollama Adapter.
There are three ways to configure Instructor to use Ollama,
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.Ollama, ollama: [...]]])
	via config :instructor, adapter: Instructor.Adapters.Ollama, ollama: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime
config = [
  adapter: Instructor.Adapters.Ollama
]

defmodule President do
  use Ecto.Schema
  use Instructor

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    model: "qwen2.5:7b",
    mode: :json,
    response_model: President,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)
There are three ways to configure Instructor to use Ollama,
	via Mix.install([...], config)
	via config :instructor, adapter: :ollama, ollama: config
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime



  

  
    
    Local Instructor w/ vLLM - Instructor v0.1.0
    
    

    



  
  

    
Local Instructor w/ vLLM
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
  ]
)

  
    
  
  Introduction


When it comes to local inference, there are three main tools that people use: olama, llama.cpp, and in professional settings, vllm.  There are obviously more, but these are the main ones that people use.  vLLM is a high performance inference server that can handle many concurrent requests in parallel.  It uses this feature called grouped attention to make sure it can utilize the GPU's VRAM very efficiently.  On an RTX 4090, you can expect the LLM to push ~3000 tokens/sec on a 7B parameter model.
vLLM is a great option if you want to host a local LLM inference server on an old gaming machine.  Or, in a corporate environment, you can host this on a GPU-optimized EC2 instance for completely private inference.
To install vLLM, head on over to the docks and run through their quick start guide.
Once installed, you can start up an OpenAI compliant inference server by running the following command,
$ vllm serve Qwen/Qwen2.5-1.5B-Instruct

This will download the model and start up your inference server. Instructor will plug in seamlessly when using the vLLM adapter.
There are three ways to configure Instructor to use vLLM,
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.VLLM, vllm: [...]]])
	via config :instructor, adapter: Instructor.Adapters.VLLM, vllm: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime
config = [
  adapter: Instructor.Adapters.VLLM
]

defmodule President do
  use Ecto.Schema
  use Instructor

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    model: "Qwen/Qwen2.5-1.5B-Instruct",
    mode: :json_schema,
    response_model: President,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)

Gemini
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
  ]
)

  
    
  
  Introduction


Google is in the LLM game, i'm sure you've heard. Currently not many people use them, however that's not for lack of quality or for absurding pricing. Actually, they're quite cheap and incredibly good for the price. The main reason why people don't use them is that they don't use an OpenAI-compliant API. Therefore most of the libraries don't support it. Not instructor, no. We support it. We have to take some liberties to ensure that the API conforms, but it works and you can be up and running before you know it.
First, head on over to the Google AI Studio to get an API key.
There are three ways to configure Instructor to use Gemini.
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.Gemini, gemini: [...]]])
	via config :instructor, adapter: Instructor.Adapters.Gemini, gemini: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime.
We recommend that you use Instructor with mode: :json_schema.
config = [
  adapter: Instructor.Adapters.Gemini,
  api_key: System.fetch_env!("LB_GEMINI_API_KEY")
]

defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    model: "gemini-2.0-flash",
    mode: :json_schema,
    response_model: President,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)
And there you have it. Instructor with Google Gemini.

Anthropic
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
  ]
)

  
    
  
  Introduction


Anthropic is a leading AI lab that competes with both OpenAI and Google.  Their models rank among the top, though their pricing is a little more expensive.  Lucky for you, we support them out of the box.  All you need to do is create an account with Anthropic and get an API key.
There are three ways to configure Instructor to use Anthropic.
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.Anthropic, anthropic: [...]]])
	via config :instructor, adapter: Instructor.Adapters.Anthropic, anthropic: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime
config = [
  adapter: Instructor.Adapters.Anthropic,
  api_key: System.fetch_env!("LB_ANTHROPIC_API_KEY")
]

defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    model: "claude-3-5-haiku-20241022",
    mode: :tools,
    max_tokens: 1000,
    response_model: President,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)
To get a list of all the models supported by Anthropic, you can check out this link.

Groq
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
  ]
)

  
    
  
  Introduction


Groq is an LLM provider that uses custom chips to provide considerably higher tokens/sec for open source models than your standard AI labs using Nvidia hardware. That being said they currently only run OSS models and they tend to lag in capability from the frontier models by 6-8 months. Groq is a great option if you want speed and good price.
To get started make an account with Groq and get an API key.
There are three ways to configure Instructor to use Groq.
	via Mix.install([...], [instructor: [adapter: Instructor.Adapters.Groq, groq: [...]]])
	via config :instructor, adapter: Instructor.Adapters.Groq, groq: [...]
	At runtime via Instructor.chat_completion(..., config)

For brevity, in this livebook, we'll configure it at runtime.
config = [
  adapter: Instructor.Adapters.Groq,
  api_key: System.fetch_env!("LB_GROQ_API_KEY"),
]

defmodule President do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)
    field(:entered_office_date, :date)
  end
end

Instructor.chat_completion(
  [
    model: "llama-3.3-70b-versatile",
    mode: :tools,
    response_model: President,
    messages: [
      %{role: "user", content: "Who was the first president of the United States?"}
    ]
  ],
  config
)
And there you go, Instructor running at 275 tokens/second.

Azure OpenAI
    

Configure your project like so to issue requests against Azure OpenAI, according to the docs.
azure_openai_endpoint = "https://contoso.openai.azure.com"
azure_openai_deployment_name = "contosodeployment123"
azure_openai_api_path = "/openai/deployments/#{azure_openai_deployment_name}/chat/completions?api-version=2024-02-01"
The Azure OpenAI service supports two authentication methods, API key and Entra ID. API key-based authN is conveyed in the api-key HTTP header, while Entra ID-issued access tokens go into the Authorization: Bearer header:

  
    
  
  API Key Authentication


config: [
  instructor: [
    adapter: Instructor.Adapters.OpenAI,
    openai: [
      auth_mode: :api_key_header, 
      api_key: System.get_env("LB_AZURE_OPENAI_API_KEY"), # e.g. "c3829729deadbeef382938acdfee2987"
      api_url: azure_openai_endpoint,
      api_path:azure_openai_api_path
    ]
  ]
]

  
    
  
  Microsoft Entra ID authentication


The code below contains a simple GenServer that continuously refreshes the access token for a service principal. Instead of setting the configuration to a fixed access token (that would expire after an hour), the api_key is set to a /0-arity function that returns the most recently fetched access token.
defmodule AzureServicePrincipalTokenRefresher do
  use GenServer

  @derive {Inspect,
           only: [:tenant_id, :client_id, :scope, :error], except: [:client_secret, :access_token]}
  @enforce_keys [:tenant_id, :client_id, :client_secret, :scope]
  defstruct [:tenant_id, :client_id, :client_secret, :scope, :access_token, :error]

  def get_token_func!(tenant_id, client_id, client_secret, scope) do
    {:ok, pid} = __MODULE__.start_link(tenant_id, client_id, client_secret, scope)

    fn ->
      case __MODULE__.get_access_token(pid) do
        {:ok, access_token} -> access_token
        {:error, error} -> raise "Could not fetch Microsoft Entra ID token: #{inspect(error)}"
      end
    end
  end

  def start_link(tenant_id, client_id, client_secret, scope) do
    GenServer.start_link(__MODULE__, %__MODULE__{
      tenant_id: tenant_id,
      client_id: client_id,
      client_secret: client_secret,
      scope: scope
    })
  end

  def get_access_token(pid) do
    GenServer.call(pid, :get_access_token)
  end

  @impl GenServer
  def init(%__MODULE__{} = state) do
    {:ok, state, {:continue, :fetch_token}}
  end

  @impl GenServer
  def handle_call(:get_access_token, _from, %__MODULE__{} = state) do
    case state do
      %__MODULE__{access_token: access_token, error: nil} ->
        {:reply, {:ok, access_token}, state}

      %__MODULE__{access_token: nil, error: error} ->
        {:reply, {:error, error}, state}
    end
  end

  @impl GenServer
  def handle_continue(:fetch_token, %__MODULE__{} = state) do
    {:noreply, fetch_token(state)}
  end

  @impl GenServer
  def handle_info(:refresh_token, %__MODULE__{} = state) do
    {:noreply, fetch_token(state)}
  end

  defp fetch_token(%__MODULE__{} = state) do
    %__MODULE__{
      tenant_id: tenant_id,
      client_id: client_id,
      client_secret: client_secret,
      scope: scope
    } = state

    case Req.post(
           url: "https://login.microsoftonline.com/#{tenant_id}/oauth2/v2.0/token",
           form: [
             grant_type: "client_credentials",
             scope: scope,
             client_id: client_id,
             client_secret: client_secret
           ]
         ) do
      {:ok,
       %Req.Response{
         status: 200,
         body: %{
           "access_token" => access_token,
           "expires_in" => expires_in
         }
       }} ->
        fetch_new_token_timeout = to_timeout(%Duration{second: expires_in - 60})
        Process.send_after(self(), :refresh_token, fetch_new_token_timeout)
        %__MODULE__{state | access_token: access_token, error: nil}

      {:ok, response} ->
        %__MODULE__{state | access_token: nil, error: response}

      {:error, error} ->
        %__MODULE__{state | access_token: nil, error: error}
    end
  end
end
Then use the helper class to configure the dynamic credential:
config: [
  instructor: [
    adapter: Instructor.Adapters.OpenAI,
    openai: [
      auth_mode: :bearer, 
      api_key: AzureServicePrincipalTokenRefresher.get_token_func!(
          System.get_env("LB_AZURE_ENTRA_TENANT_ID"), # e.g. "contoso.onmicrosoft.com"
          System.get_env("LB_AZURE_OPENAI_CLIENT_ID"), # e.g. "deadbeef-0000-4f13-afa9-c8a1e4087f97"
          System.get_env("LB_AZURE_OPENAI_CLIENT_SECRET"), # e.g. "mEf8Q~.e2e8URInwinsermNe8wDewsedRitsen.."}, 
          "https://cognitiveservices.azure.com/.default"
      ),  
      api_url: azure_openai_endpoint,
      api_path: azure_openai_api_path
    ]
  ]
]



  

  
    
    Text Classification - Instructor v0.1.0
    
    

    



  
  

    
Text Classification
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


Text classification is a common task in NLP and broadly applicable across software. Whether it be spam detection, or support ticket categorization, NLP is at the core. Historically, this required training custom, bespoke models that required collecting thousands of pre-labeled examples. With LLMs a lot of this knowledge is already encoded into the model. With proper instruction and guiding the output to a known set of classifications using GPT you can be up and running with a text classification model in no time.
Hell, you can even use instructor to help generate the training set to train your own more efficient model. But let's not get ahead of ourselves, there's more on that later in the tutorials.

  
    
  
  Binary Text Classification


Spam detection is a classic example of binary text classification. It's as simple as returning a true / false of whether an example is in the class. This is pretty trivial to implement in instructor.
defmodule SpamPrediction do
  use Ecto.Schema
  use Instructor

  @llm_doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam.
  - reason: A short, less than 10 word rationalization for the classification.
  - score: A confidence score between 0.0 and 1.0 for the classification.
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    response_model: SpamPrediction,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        Your purpose is to classify customer support emails as either spam or not.
        This is for a clothing retail business.
        They sell all types of clothing.

        Classify the following email: 
        ```
        #{text}
        ```
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")
{:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince scam email", score: 0.95}}
We don't have to stop just at a boolean inclusion, we can also easily extend this idea to multiple categories or classes that we can classify the text into. In this example, let's consider classifying support emails. We want to know whether it's a general_inquiry, billing_issue, or a technical_issue perhaps it rightly fits in multiple classes. This can be useful if we want to cc' specialized support agents when intersecting customer issues occur
We can leverage Ecto.Enum to define a schema that restricts the LLM output to be a list of those values. We can also provide a @doc description to help guide the LLM with the semantic understanding of what these classifications ought to represent.
defmodule EmailClassifications do
  use Ecto.Schema
  use Instructor

  @llm_doc """
  A classification of a customer support email.

  technical_issue - whether the user is having trouble accessing their account.
  billing_issue - whether the customer is having trouble managing their billing or credit card
  general_inquiry - all other issues
  """
  @primary_key false
  embedded_schema do
    field(:tags, {:array, Ecto.Enum},
      values: [:general_inquiry, :billing_issue, :technical_issue]
    )
  end
end

classify_email = fn text ->
  {:ok, %{tags: result}} =
    Instructor.chat_completion(
      model: "gpt-4o-mini",
      response_model: EmailClassifications,
      messages: [
        %{
          role: "user",
          content: "Classify the following text: #{text}"
        }
      ]
    )

  result
end

classify_email.("My account is locked and I can't access my billing info.")
[:technical_issue, :billing_issue]

Extracting Action Items from Meeting Transcripts
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino, "~> 0.12.0"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


This example shouldn't be foreign to any of you. I'm sure each day you log in, join a Zoom call, and have a meeting about what y'all are going to do this upcoming week. It is then the work of some product manager to translate this into a JIRA board so that you can track your progress throughout the week.
The bane of most engineers existence... We can automate this.

  
    
  
  The Schema


Let's start by defining a schema for the tickets and the subtasks that might exist within them. There will be a priority, description and a set of dependencies between the tasks for the tickets.
defmodule MeetingNotes do
  use Ecto.Schema
  use Instructor.Validator

  @llm_doc """
  Tickets correctly resolved from a meeting transcription.

  Tickets have a name, priority, useful description, and assignees.
  They may also have subtasks that share the same id spaces as the tickets.
  A ticket also may have references to Tickets and Subtasks that are a blocking dependency
  for completing the ticket.
  """
  @primary_key false
  embedded_schema do
    embeds_many :tickets, Ticket do
      field(:name, :string)
      field(:description, :string)
      field(:priority, Ecto.Enum, values: [:high, :medium, :low])
      field(:assignees, {:array, :string})

      embeds_many :subtasks, SubTasks do
        field(:name, :string)
      end

      field(:dependencies, {:array, :binary_id})
    end
  end
end
warning: module attribute @llm_doc was set but never used
└─ pages/cookbook/extract-action-items-from-meeting-transcripts.livemd#cell:2eli2kgfgvm2ligb:5: MeetingNotes (module)

{:module, MeetingNotes, <<70, 79, 82, 49, 0, 0, 41, ...>>, :ok}
One thing interesting to note here is that although we could use proptorecto associations, in this case it's rather tedious. Since our dependencies can either be on tickets or on subtasks, you would typically have to have a polymorphic relationship.
While Instructor fully supports doing that, we can instead just create embedded schemas for the subtasks and define that the subtasks and the tickets share the same ID space in our doc comment. This will steer the LLM to produce foreign key relations in a polymorphic way.
One also might choose to solve this modeling problem by denormalizing the tickets and subtasks into its own task type and get the LLM to not only have the dependencies listed but also any subtasks that a ticket might have through association instead of embedding.
Either way works, but in this example we prefer the embedded method because it produces a prettier output and we can lean on the LLM a little more.
Now, let's extract some JIRA tickets.
generate_tickets = fn transcript ->
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    response_model: MeetingNotes,
    messages: [
      %{role: "system", content: "The following is a transcript of a meeting..."},
      %{
        role: "user",
        content: "Create the action items for the following transcript: #{transcript}"
      }
    ]
  )
end

{:ok, %{tickets: tickets}} =
  generate_tickets.("""
    Alice: Hey team, we have several critical tasks we need to tackle for the upcoming release. First, we need to work on improving the authentication system. It's a top priority.
    
    Bob: Got it, Alice. I can take the lead on the authentication improvements. Are there any specific areas you want me to focus on?
    
    Alice: Good question, Bob. We need both a front-end revamp and back-end optimization. So basically, two sub-tasks.
    
    Carol: I can help with the front-end part of the authentication system.
    
    Bob: Great, Carol. I'll handle the back-end optimization then.
    
    Alice: Perfect. Now, after the authentication system is improved, we have to integrate it with our new billing system. That's a medium priority task.
    
    Carol: Is the new billing system already in place?
    
    Alice: No, it's actually another task. So it's a dependency for the integration task. Bob, can you also handle the billing system?
    
    Bob: Sure, but I'll need to complete the back-end optimization of the authentication system first, so it's dependent on that.
    
    Alice: Understood. Lastly, we also need to update our user documentation to reflect all these changes. It's a low-priority task but still important.
    
    Carol: I can take that on once the front-end changes for the authentication system are done. So, it would be dependent on that.
    
    Alice: Sounds like a plan. Let's get these tasks modeled out and get started.
  """)
{:ok,
 %MeetingNotes{
   tickets: [
     %MeetingNotes.Ticket{
       id: "1",
       name: "Improve authentication system",
       description: "Work on improving the authentication system",
       priority: :high,
       assignees: ["Bob"],
       subtasks: [
         %MeetingNotes.Ticket.SubTasks{id: "1", name: "Front-end revamp"},
         %MeetingNotes.Ticket.SubTasks{id: "2", name: "Back-end optimization"}
       ],
       dependencies: []
     },
     %MeetingNotes.Ticket{
       id: "2",
       name: "Integrate authentication system with new billing system",
       description: "Integrate the improved authentication system with the new billing system",
       priority: :medium,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: ["1"]
     },
     %MeetingNotes.Ticket{
       id: "3",
       name: "Update user documentation",
       description: "Update user documentation to reflect changes",
       priority: :low,
       assignees: ["Carol"],
       subtasks: [],
       dependencies: ["1.1"]
     }
   ]
 }}
The results look good. We can use Kino to then render out all the dependencies of our tickets.
generate_tickets_diagram = fn tickets ->
  subtasks = Enum.flat_map(tickets, & &1.subtasks)
  all_tasks = subtasks ++ tickets

  ticket_nodes =
    tickets
    |> Enum.map_join("\n", fn t ->
      subtask_relations =
        t.subtasks
        |> Enum.map_join("\n", fn st ->
          """
          "#{t.name}" ||--o| "#{st.name}" : "Has Subtask"
          """
        end)

      dependency_relations =
        t.dependencies
        |> Enum.map_join("\n", fn d ->
          dt = all_tasks |> Enum.find(&(&1.id == d))

          if dt do
            """
            "#{t.name}" ||--o| "#{dt.name}" : "Depends on"
            """
          else
            ""
          end
        end)

      """
        "#{t.name}" {
          priority #{t.priority}
          assignees #{Enum.join(t.assignees, ", ")}
        }

        #{subtask_relations}
        #{dependency_relations}
      """
    end)

  subtask_nodes =
    tickets
    |> Enum.flat_map(& &1.subtasks)
    |> Enum.map_join("\n", fn st ->
      """
        "#{st.name}"
      """
    end)

  Kino.Mermaid.new("""
    erDiagram
      #{ticket_nodes}
      #{subtask_nodes}
  """)
end

generate_tickets_diagram.(tickets)
  erDiagram
      "Improve authentication system" {
    priority high
    assignees Bob
  }

  "Improve authentication system" ||--o| "Front-end revamp" : "Has Subtask"

"Improve authentication system" ||--o| "Back-end optimization" : "Has Subtask"

  

  "Integrate authentication system with new billing system" {
    priority medium
    assignees Bob
  }

  
  "Integrate authentication system with new billing system" ||--o| "Front-end revamp" : "Depends on"


  "Update user documentation" {
    priority low
    assignees Carol
  }

  
  

      "Front-end revamp"

  "Back-end optimization"





  

  
    
    Converting Text into Explorer DataFrames - Instructor v0.1.0
    
    

    



  
  

    
Converting Text into Explorer DataFrames
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:explorer, "~> 0.7.2"},
    {:kino, "~> 0.12.0"},
    {:kino_explorer, "~> 0.1.13"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Motivation


Sometimes we come across text that can be naturally extracted into structured data. Although we don't necessarily know how many tables, and what the schemas are represented in the data. Using Instructor we can let the LLM determine the tables and schemas for us, pulling the results into a dataframe which we can then use for further analysis.

  
    
  
  The Schema


Notice from this example we want resulting data structure to have data frames. That's not a default Ecto type so we're gonna have to do things a little different here.
Luckily, Ecto allows us to define our own custom types in which we can override how the values are casted. And Instructor provides us a behavior which tells us how to represent it in a JSONSchema so the LLM can understand it.
In our case, we're going to create a type for a table of data, which will be represented as a CSV string with respect to the LLM. Our custom type will then parse the string when it is casted by Ecto and parse the CSV into an Explorer.DataFrame.
defmodule Ecto.CSVDataFrame do
  use Ecto.Type
  use Instructor.EctoType

  def type, do: :string

  def cast(csv_str) when is_binary(csv_str) do
    df = Explorer.DataFrame.load_csv!(csv_str)
    {:ok, df}
  end

  def cast(%Explorer.DataFrame{} = df), do: {:ok, df}
  def cast(_), do: :error

  def to_json_schema(),
    do: %{
      type: "string",
      description: "A CSV representation of a data table"
    }

  def dump(x), do: {:ok, x}
  def load(x), do: {:ok, x}
end
{:module, Ecto.CSVDataFrame, <<70, 79, 82, 49, 0, 0, 17, ...>>, {:load, 1}}
Now that we have our data type, we can use it in our embedded ecto schema, just as we would have in any other example. If everything works correctly, we should get some data frames out.
defmodule Database do
  use Ecto.Schema
  use Instructor

  @llm_doc """
  The extracted database will contain one or more tables with data as csv formatted with ',' delimiters
  """
  @primary_key false
  embedded_schema do
    embeds_many :tables, DataFrame, primary_key: false do
      field(:name, :string)
      field(:data, Ecto.CSVDataFrame)
    end
  end
end
{:module, Database, <<70, 79, 82, 49, 0, 0, 43, ...>>, :ok}
Perfect. And now let's test it out on some example text.
extract_df = fn text ->
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    response_model: Database,
    messages: [
      %{
        role: "system",
        content: """
        Map this data into one or more dataframes and correctly define the columns and data
        """
      },
      %{
        role: "user",
        content: "#{text}"
      }
    ]
  )
end

{:ok, db} =
  extract_df.("""
    My name is John and I am 25 years old. I live in
    New York and I like to play basketball. His name is
    Mike and he is 30 years old. He lives in San Francisco
    and he likes to play baseball. Sarah is 20 years old
    and she lives in Los Angeles. She likes to play tennis.
    Her name is Mary and she is 35 years old.
    She lives in Chicago.
    
    On one team 'Tigers' the captain is John and there are 12 players.
    On the other team 'Lions' the captain is Mike and there are 10 players.
  """)

Kino.Layout.tabs(
  for table <- db.tables do
    {table.name, table.data}
  end
)

GPT-4o-vision - Extracting Data from Images
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino, "~> 0.12.3"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [
        api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      ]
    ]
  ]
)

  
    
  
  Motivation


The latest models support vision capabilities as well. This, with no extra work, is a feature of Instructor. All you have to do is pass a URL or Base64 encoded image as one of the messages, and everything should just work seamlessly.
In the following example, we will extract product details from a screenshot of a Shopify store.
[image: ]
image = Kino.FS.file_path("shopify-screenshot.png") |> File.read!()
base64_image = "data:image/png;base64," <> Base.encode64(image)

defmodule Product do
  use Ecto.Schema

  @primary_key false
  embedded_schema do
    field(:name, :string)
    field(:price, :decimal)
    field(:currency, Ecto.Enum, values: [:usd, :gbp, :eur, :cny])
    field(:color, :string)
  end
end

{:ok, result} =
  Instructor.chat_completion(
    model: "gpt-4o",
    response_model: Product,
    messages: [
      %{
        role: "user",
        content: [
          %{type: "text", text: "What is the product details of the following image?"},
          %{type: "image_url", image_url: %{url: base64_image, detail: "high"}}
        ]
      }
    ]
  )

result
%Product{
  name: "Thomas Wooden Railway Thomas The Tank Engine",
  price: Decimal.new("33.0"),
  currency: :usd,
  color: "blue"
}

Extracting data from images of Receipts
    

Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)},
    {:kino, "~> 0.12.3"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [
        api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      ]
    ]
  ]
)

  
    
  
  Our Object Model


image = Kino.FS.file_path("receipt.jpg") |> File.read!()
<<255, 216, 255, 224, 0, 16, 74, 70, 73, 70, 0, 1, 1, 0, 0, 1, 0, 1, 0, 0, 255, 226, 2, 40, 73, 67,
  67, 95, 80, 82, 79, 70, 73, 76, 69, 0, 1, 1, 0, 0, 2, 24, 0, 0, 0, 0, 2, 16, 0, 0, ...>>
We can look at the image of the receipt and outline the fields that we want to extract using an Ecto schema, as shown below.
defmodule Receipt do
  use Ecto.Schema
  use Instructor.Validator
  
  import Ecto.Changeset

  @primary_key false
  embedded_schema do
    field(:total, :decimal)
    field(:subtotal, :decimal)

    embeds_many :items, Item do
      field(:name, :string)
      field(:price, :decimal)
      field(:quantity, :integer)
    end
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> validate_required([:total, :subtotal])
    |> validate_items_total()
  end

  defp validate_items_total(changeset) do
    items = get_field(changeset, :items) || []
    subtotal = get_field(changeset, :subtotal)

    items_total = Enum.reduce(items, Decimal.new(0), fn item, acc ->
      item_total = Decimal.mult(item.price, Decimal.new(item.quantity))
      Decimal.add(acc, item_total)
    end)

    if Decimal.equal?(items_total, subtotal) do
      changeset
    else
      add_error(changeset, :subtotal, """
        Subtotal does not match sum of item prices. 
        Subtotal is #{subtotal} and items total is #{items_total}")
      """)
    end
  end
end

Kino.nothing()

  
    
  
  Validating the Subtotal


Notice how we used the validate_changeset callback to check that the items we extract sum up to the subtotal on the receipt. This process allows us to gain confidence that the OCR application is functioning correctly. In fact, later on, we can even use the re-ask feature of the instructor to have the LLM correct any errors itself.

  
    
  
  Calling the LLM


base64_image = "data:image/jpeg;base64," <> Base.encode64(image)

{:ok, receipt} = Instructor.chat_completion(
  model: "gpt-4o",
  response_model: Receipt,
  messages: [
    %{
      role: "user",
      content: [
        %{
          type: "image_url",
          image_url: %{url: base64_image}
        },
        %{
          type: "text",
          text: "Analyze the image and return the items in the receipt and the total amount."
        }
      ]
    }
  ]
)

Kino.Layout.grid([Kino.Image.new(image, :jpeg), receipt], columns: 2, boxed: true)
%Receipt{
  total: Decimal.new("107.6"),
  subtotal: Decimal.new("107.6"),
  items: [
    %Receipt.Item{id: "1", name: "Lorem ipsum", price: Decimal.new("9.2"), quantity: 1},
    %Receipt.Item{id: "2", name: "Lorem ipsum dolor sit", price: Decimal.new("19.2"), quantity: 1},
    %Receipt.Item{
      id: "3",
      name: "Lorem ipsum dolor sit amet",
      price: Decimal.new("15.0"),
      quantity: 1
    },
    %Receipt.Item{id: "4", name: "Lorem ipsum", price: Decimal.new("15.0"), quantity: 1},
    %Receipt.Item{id: "5", name: "Lorem ipsum", price: Decimal.new("15.0"), quantity: 1},
    %Receipt.Item{id: "6", name: "Lorem ipsum dolor sit", price: Decimal.new("15.0"), quantity: 1},
    %Receipt.Item{id: "7", name: "Lorem ipsum", price: Decimal.new("19.2"), quantity: 1}
  ]
}
Now, we simply using Instructor to call gpt-4o with the base64 encoded image and the response model, and we can get back the results. We can have confidence that the results returned match all of our validations, and we've reduced the effects of any hallucinations.

PII Data Sanitization
    


  
    
  
  Section


Mix.install(
  [
    {:instructor, path: Path.expand("../../", __DIR__)}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI,
      openai: [api_key: System.fetch_env!("LB_OPENAI_API_KEY")]
    ]
  ]
)

  
    
  
  Overview


This example demonstrates the usage of OpenAI's ChatCompletion model for the extraction and scrubbing of Personally Identifiable Information (PII) from an input. The code defines Ecto schema to manage the PII data and offers function for both extraction and sanitation.
Defining the Structures
First, Ecto schemas are defined to represent the PII data and the overall structure for PII data extraction.
defmodule PII do
  use Ecto.Schema
  use Instructor

  @llm_doc """
  ## Field Descriptions:
  - index: an auto incrementing integer starting at zero
  - type: the type of personal identifiable information
  - value: the PII value
  """
  @primary_key false
  embedded_schema do
    embeds_many :data, Datum, primary_key: false do
      field(:index, :integer)
      field(:type, :string)
      field(:value, :string)
    end
  end

  @doc """
  Iterates over the private data and replaces the value with a placeholder in the
  form of <{data_type}_{i}>
  """
  def scrub({:ok, pii}, input) do
    Enum.reduce(pii.data, input, fn datum, acc ->
      String.replace(acc, datum.value, "<#{datum.type}_#{datum.index}>")
    end)
  end

  def scrub({:error, reason}, _input) do
    dbg(reason)
  end

  def extract(input) do
    Instructor.chat_completion(
      model: "gpt-4o-mini",
      response_model: PII,
      max_retries: 3,
      messages: [
        %{
          role: "system",
          content:
            "You are a world class PII scrubbing model, Extract the PII data from the following document"
        },
        %{
          role: "system",
          content: """
          Examples of PII: names, addresses, phone numbers, email addresses, financial information
          """
        },
        %{
          role: "system",
          content: """
          Instructions:
          - any spaces in the type should be converted to underscores and all letters should be lower case
          - use abbreviations when choosing the type
          """
        },
        %{
          role: "user",
          content: input
        }
      ]
    )
  end
end

  
    
  
  Extracting PII Data


The OpenAI API is utilized to extract PII information from a given input.
input =
  "Hello John Smith, I am Jill. Your GitBoat, LLC credit card account 1111-0000-1111-8765 has a minimum payment of $33.32 that is due by July 24th."

pii_data = PII.extract(input)

  
    
  
  Scrubbing PII Data


After extracting the PII data, the PII.scrub/2 funnction is used to sanitize the input.
PII.scrub(pii_data, input)

Instructor 
    



      
Check out our Quickstart Guide to get up and running with Instructor in minutes.
Instructor provides structured prompting for LLMs. It is a spiritual port of the great Instructor Python Library by @jxnlco.
Instructor allows you to get structured output out of an LLM using Ecto.
You don't have to define any JSON schemas.
You can just use Ecto as you've always used it.
And since it's just ecto, you can provide change set validations that you can use to ensure that what you're getting back from the LLM is not only properly structured, but semantically correct.
To learn more about the philosophy behind Instructor and its motivations, check out this Elixir Denver Meetup talk:

[![Instructor: Structured prompting for LLMs](assets/youtube-thumbnail.png)](https://www.youtube.com/watch?v=RABXu7zqnT0)

While Instructor is designed to be used with OpenAI, it also supports every major AI lab and open source LLM inference server:
	OpenAI
	Anthropic
	Groq
	Ollama
	Gemini
	vLLM
	llama.cpp

At its simplest, usage is pretty straightforward: 
	Create an ecto schema, with a @llm_doc string that explains the schema definition to the LLM. 
	Define a validate_changeset/1 function on the schema, and use the use Instructor macro in order for Instructor to know about it.
	Make a call to Instructor.chat_completion/1 with an instruction for the LLM to execute.

You can use the max_retries parameter to automatically, iteratively go back and forth with the LLM to try fixing validation errorswhen they occur.
Mix.install([:instructor])

defmodule SpamPrediction do
  use Ecto.Schema
  use Validator

  @llm_doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam.
  - reason: A short, less than 10 word rationalization for the classification.
  - score: A confidence score between 0.0 and 1.0 for the classification.
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-4o-mini",
    response_model: SpamPrediction,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        Your purpose is to classify customer support emails as either spam or not.
        This is for a clothing retail business.
        They sell all types of clothing.

        Classify the following email: 

        <email>
          #{text}
        </email>
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")

# => {:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince email scam", score: 0.98}}

      


      
        Summary


  
    Types
  


    
      
        stream()

      


    





  
    Functions
  


    
      
        cast_all(schema, params)

      


        Casts all the parameters in the params map to the types defined in the types map.
This works both with Ecto Schemas and maps of Ecto types (see Schemaless Ecto).



    


    
      
        chat_completion(params, config \\ nil)

      


        Create a new chat completion for the provided messages and parameters.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    stream()


      
       
       View Source
     


  


  

      

          @type stream() :: Enumerable.t()


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    cast_all(schema, params)


      
       
       View Source
     


  


  

Casts all the parameters in the params map to the types defined in the types map.
This works both with Ecto Schemas and maps of Ecto types (see Schemaless Ecto).

  
    
  
  Examples


When using a full Ecto Schema
iex> Instructor.cast_all(%{
...>   data: %Instructor.Demos.SpamPrediction{},
...>   types: %{
...>     class: :string,
...>     score: :float
...>   }
...> }, %{
...>   class: "spam",
...>   score: 0.999
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    class: "spam",
    score: 0.999
  },
  errors: [],
  data: %Instructor.Demos.SpamPrediction{
    class: :spam,
    score: 0.999
  },
  valid?: true
}
When using a map of Ecto types
iex> Instructor.cast_all(%Instructor.Demo.SpamPrediction{}, %{
...>   class: "spam",
...>   score: 0.999
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    class: "spam",
    score: 0.999
  },
  errors: [],
  data: %{
    class: :spam,
    score: 0.999
  },
  valid?: true
}
and when using raw Ecto types,
iex> Instructor.cast_all({%{},%{name: :string}, %{
...>   name: "George Washington"
...> })
%Ecto.Changeset{
  action: nil,
  changes: %{
    name: "George Washington",
  },
  errors: [],
  data: %{
    name: "George Washington",
  },
  valid?: true
}

  



    

  
    
      
      Link to this function
    
    chat_completion(params, config \\ nil)


      
       
       View Source
     


  


  

      

          @spec chat_completion(Keyword.t(), any()) ::
  {:ok, Ecto.Schema.t()}
  | {:error, Ecto.Changeset.t()}
  | {:error, String.t()}
  | stream()


      


Create a new chat completion for the provided messages and parameters.
The parameters are passed directly to the LLM adapter.
By default they shadow the OpenAI API parameters.
For more information on the parameters, see the OpenAI API docs.
Additionally, the following parameters are supported:
	:adapter - The adapter to use for chat completion. (defaults to the configured adapter, which defaults to Instructor.Adapters.OpenAI)
	:response_model - The Ecto schema to validate the response against, or a valid map of Ecto types (see Schemaless Ecto).
	:stream - Whether to stream the response or not. (defaults to false)
	:validation_context - The validation context to use when validating the response. (defaults to %{})
	:mode - The mode to use when parsing the response, :tools, :json, :md_json (defaults to :tools), generally speaking you don't need to change this unless you are not using OpenAI.
	:max_retries - The maximum number of times to retry the LLM call if it fails, or does not pass validations.             (defaults to `0`)



  
    
  
  Examples


iex> Instructor.chat_completion(
...>   model: "gpt-4o-mini",
...>   response_model: Instructor.Demos.SpamPrediction,
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
...>     }
...>   ])
{:ok,
    %Instructor.Demos.SpamPrediction{
        class: :spam
        score: 0.999
    }}
When you're using Instructor in Streaming Mode, instead of returning back a tuple, it will return back a stream that emits tuples.
There are two main streaming modes available. array streaming and partial streaming.
Partial streaming will emit the record multiple times until it's complete.
iex> Instructor.chat_completion(
...>   model: "gpt-4o-mini",
...>   response_model: {:partial, %{name: :string, birth_date: :date}}
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Who is the first president of the United States?"
...>     }
...>   ]) |> Enum.to_list()
[
  {:partial, %{name: "George Washington"}},
  {:partial, %{name: "George Washington", birth_date: ~D[1732-02-22]}},
  {:ok, %{name: "George Washington", birth_date: ~D[1732-02-22]}}
]
Whereas with array streaming, you can ask the LLM to return multiple instances of your Ecto schema,
and instructor will emit them one at a time as they arrive in complete form and validated.
iex> Instructor.chat_completion(
...>   model: "gpt-4o-mini",
...>   response_model: {:array, %{name: :string, birth_date: :date}}
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Who are the first 5 presidents of the United States?"
...>     }
...>   ]) |> Enum.to_list()

[
  {:ok, %{name: "George Washington", birth_date: ~D[1732-02-22]}},
  {:ok, %{name: "John Adams", birth_date: ~D[1735-10-30]}},
  {:ok, %{name: "Thomas Jefferson", birth_date: ~D[1743-04-13]}},
  {:ok, %{name: "James Madison", birth_date: ~D[1751-03-16]}},
  {:ok, %{name: "James Monroe", birth_date: ~D[1758-04-28]}}
]
If there's a validation error, it will return an error tuple with the change set describing the errors.
iex> Instructor.chat_completion(
...>   model: "gpt-4o-mini",
...>   response_model: Instructor.Demos.SpamPrediction,
...>   messages: [
...>     %{
...>       role: "user",
...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
...>     }
...>   ])
{:error,
    %Ecto.Changeset{
        changes: %{
            class: "foobar",
            score: -10.999
        },
        errors: [
            class: {"is invalid", [type: :string, validation: :cast]}
        ],
        valid?: false
    }}

Instructor.Adapter behaviour
    



      
Behavior for Instructor.Adapter.

      


      
        Summary


  
    Types
  


    
      
        config()

      


    


    
      
        params()

      


    


    
      
        raw_response()

      


    


    
      
        stream()

      


    





  
    Callbacks
  


    
      
        chat_completion(params, config)

      


    


    
      
        reask_messages(raw_response, params, config)

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    config()


      
       
       View Source
     


  


  

      

          @type config() :: any()


      



  



  
    
      
      Link to this type
    
    params()


      
       
       View Source
     


  


  

      

          @type params() :: [Keyword.t()]


      



  



  
    
      
      Link to this type
    
    raw_response()


      
       
       View Source
     


  


  

      

          @type raw_response() :: any()


      



  



  
    
      
      Link to this type
    
    stream()


      
       
       View Source
     


  


  

      

          @type stream() :: Enumerable.t()


      



  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    chat_completion(params, config)


      
       
       View Source
     


  


  

      

          @callback chat_completion(params(), config()) ::
  stream() | {:ok, raw_response(), String.t()} | {:error, String.t()}


      



  



  
    
      
      Link to this callback
    
    reask_messages(raw_response, params, config)


      
       
       View Source
     


  


  

      

          @callback reask_messages(raw_response(), params(), config()) :: [map()]

Instructor.Adapters.Anthropic 
    



      
Anthropic adapter for Instructor.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.Anthropic, anthropic: [
  api_key: "your_api_key" # Will use ANTHROPIC_API_KEY environment variable if not provided
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.Anthropic,
  api_key: "your_api_key" # Will use ANTHROPIC_API_KEY environment variable if not provided
])
To get an Anthropic API key, see Anthropic.

Instructor.Adapters.Gemini 
    



      
Adapter for Google Gemini.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.Gemini, gemini: [
  api_key: "your_api_key" # Will use GOOGLE_API_KEY environment variable if not provided
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.Gemini,
  api_key: "your_api_key" # Will use GOOGLE_API_KEY environment variable if not provided
])
To get a Google API key, see Google AI Studio.

      


      
        Summary


  
    Functions
  


    
      
        chat_completion(params, user_config \\ nil)

      


        Run a completion against Google's Gemini API
Accepts OpenAI API arguments and converts to Gemini Args to perform the completion.
Defaults to JSON mode within the Gemini API



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    chat_completion(params, user_config \\ nil)


      
       
       View Source
     


  


  

Run a completion against Google's Gemini API
Accepts OpenAI API arguments and converts to Gemini Args to perform the completion.
Defaults to JSON mode within the Gemini API

Instructor.Adapters.Groq 
    



      
Adapter for Groq Cloud API.  Using the OpenAI API compatible endpoint.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.Groq, groq: [
  api_key: "your_api_key" # Will use GROQ_API_KEY environment variable if not provided
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.Groq,
  api_key: "your_api_key" # Will use GROQ_API_KEY environment variable if not provided
])
For more configurations, see Instructor.Adapters.OpenAI for more details as this adapter inherits functionality from it.
To get a Groq API key, see Groq Cloud.

Instructor.Adapters.Llamacpp 
    



      
Runs against the llama.cpp server. To be clear this calls the llamacpp specific
endpoints, not the open-ai compliant ones.
You can read more about it here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.Llamacpp, llamacpp: [
  api_url: "http://localhost:8080" # (Optional) defaults to localhost:8080
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.Llamacpp,
  api_url: "http://localhost:8080"
])

      


      
        Summary


  
    Functions
  


    
      
        chat_completion(params, config \\ nil)

      


        Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    chat_completion(params, config \\ nil)


      
       
       View Source
     


  


  

Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.
You can read more about the parameters here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

  
    
  
  Examples


  iex> Instructor.chat_completion(
  ...>   model: "llama3.1-8b-instruct",
  ...>   messages: [
  ...>     %{ role: "user", content: "Classify the following text: Hello I am a Nigerian prince and I would like to send you money!" },
  ...>   ],
  ...>   response_model: response_model,
  ...>   temperature: 0.5,
  ...> )

  


        

      



  

  
    
    Instructor.Adapters.Ollama - Instructor v0.1.0
    
    

    



  
  

    
Instructor.Adapters.Ollama 
    



      
Ollama adapter for Instructor.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.Ollama, ollama: [
  api_url: "http://localhost:11434" # (Optional) defaults to localhost:11434
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.Ollama,
  api_url: "http://localhost:11434" # (Optional) defaults to localhost:11434
])
To get an Ollama API key, see Ollama.

Instructor.Adapters.OpenAI 
    



      
Documentation for Instructor.Adapters.OpenAI.

Instructor.Adapters.VLLM 
    



      
VLLM adapter for Instructor.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.VLLM, vllm: [
  api_url: "http://localhost:8000" # (Optional) defaults to localhost:8000
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.VLLM,
  api_url: "http://localhost:8000" # (Optional) defaults to localhost:8000
])
To get a VLLM API key, see VLLM.

Instructor.Adapters.XAI 
    



      
Adapter for XAI API to access the Grok models.

  
    
  
  Configuration


config :instructor, adapter: Instructor.Adapters.XAI, xai: [
  api_key: "your_api_key" # Will use XAI_API_KEY environment variable if not provided
]
or at runtime:
Instructor.chat_completion(..., [
  adapter: Instructor.Adapters.XAI,
  api_key: "your_api_key" # Will use XAI_API_KEY environment variable if not provided
])

Instructor.EctoType behaviour
    



      
Instructor.EctoType is a behaviour that lets your implement your own custom Ecto.Type
  that works natively with Instructor.

  
    
  
  Example


  defmodule MyCustomType do
    use Ecto.Type
    use Instructor.EctoType

    # ... See `Ecto.Type` for implementation details

    def to_json_schema() do
      %{
        type: "string",
        format: "email"
      }
    end
  end

      


      
        Summary


  
    Callbacks
  


    
      
        to_json_schema()

      


    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    to_json_schema()


      
       
       View Source
     


  


  

      

          @callback to_json_schema() :: map()

Instructor.Extras.ChainOfThought 
    




      
        Summary


  
    Functions
  


    
      
        chat_completion(params, config \\ nil)

      


    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    chat_completion(params, config \\ nil)


      
       
       View Source

Instructor.Extras.ChainOfThought.ReasoningStep 
    




      
        Summary


  
    Functions
  


    
      
        %Instructor.Extras.ChainOfThought.ReasoningStep{}

      


        For each step, provide a title that describes what you're doing in that step, along with the content.
Decide if you need another step or if you're ready to give the final answer.
Respond in JSON format with 'title', 'content', and 'next_action' (either 'continue' or 'final_answer') keys.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    %Instructor.Extras.ChainOfThought.ReasoningStep{}


      
       
       View Source
     


      (struct)

  


  

For each step, provide a title that describes what you're doing in that step, along with the content.
Decide if you need another step or if you're ready to give the final answer.
Respond in JSON format with 'title', 'content', and 'next_action' (either 'continue' or 'final_answer') keys.

Instructor.JSONSchema 
    




      
        Summary


  
    Functions
  


    
      
        from_ecto_schema(ecto_schema)

      


          Generates a JSON Schema from an Ecto schema.



    


    
      
        traverse_and_update(tree, fun, opts \\ [])

      


        Traverses a tree structure of maps and lists, allowing the user to update or remove elements.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    from_ecto_schema(ecto_schema)


      
       
       View Source
     


  


  

  Generates a JSON Schema from an Ecto schema.
  Note: This will output a correct JSON Schema for the given Ecto schema, but
  it will not necessarily be optimal, nor support all Ecto types.

  



    

  
    
      
      Link to this function
    
    traverse_and_update(tree, fun, opts \\ [])


      
       
       View Source
     


  


  

Traverses a tree structure of maps and lists, allowing the user to update or remove elements.

  
    
  
  Parameters


	tree: The tree structure to traverse (can be a map, list, or any other type)
	fun: A function that takes either:	Just the element if include_path: false (default)
	A tuple of {element, path} if include_path: true, where path is a list of keys to reach this element
The function should return either:
	An updated element
	nil to remove the element
	The original element if no changes are needed


	opts: Optional keyword list of options	include_path: boolean, when true includes the path to each element in the callback (default: false)




  
    
  
  Returns


  The updated tree structure

        parse(chunks)

      Link to this function
    
    parse(chunks)


      
       
       View Source

Instructor.Validator behaviour
    



      
By default you'll get whatever OpenAI returns.
This behavior provides a hook for you to critique the response using standard ecto changesets validations.
This can be used in conjuction with the :max_retries parameter to Instructor.chat_completion/1 to retry the completion until it passes your validation.

  
    
  
  Examples


  defmodule Instructor.Demos.SpamPrediction do
use Ecto.Schema

@primary_key false
schema "spam_prediction" do
    field :class, :string
    field :score, :float
end

@impl true
def validate_changeset(changeset) do
    changeset
    |> validate_number(:score, less_than_or_equal_to: 1.0)
end
  end
  iex> Instructor.chat_completion(
  ...>   model: "gpt-4o-mini",
  ...>   response_model: Instructor.Demos.SpamPrediction,
  ...>   max_retries: 1,
  ...>   messages: [
  ...>     %{
  ...>       role: "user",
  ...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
  ...>     }
  ...>   ])
  {:error, %Ecto.Changeset{
  action: nil,
  changes: %{},
  errors: [
      score: {"is invalid", [validation: :number, validation_opts: [less_than_or_equal_to: 1.0]]}
  ],
  data: %Instructor.Demos.SpamPrediction{
      class: nil,
      score: nil
  },
  valid?: false
  }}

    Callbacks
  


    
      
        validate_changeset(t)

      


    


    
      
        validate_changeset(t, map)

      


    





  
    Functions
  


    
      
        validate_with_llm(changeset, field, statement, opts \\ [])

      


        Validate a changeset field using a language model



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    validate_changeset(t)


      
       
       View Source
     


      (optional)

  


  

      

          @callback validate_changeset(Ecto.Changeset.t()) :: Ecto.Changeset.t()


      



  



  
    
      
      Link to this callback
    
    validate_changeset(t, map)


      
       
       View Source
     


      (optional)

  


  

      

          @callback validate_changeset(Ecto.Changeset.t(), map()) :: Ecto.Changeset.t()


      



  


        

      

      
        Functions

        


    

  
    
      
      Link to this function
    
    validate_with_llm(changeset, field, statement, opts \\ [])


      
       
       View Source
     


  


  

Validate a changeset field using a language model

  
    
  
  Example


  defmodule QuestionAnswer do
use Ecto.Schema

@primary_key false
embedded_schema do
    field :question, :string
    field :answer, :string
end

@impl true
def validate_changeset(changeset) do
    changeset
    |> validate_with_llm(:answer, "do not say anything objectionable")
end
  end

